
Goal setting 

 

The Excel schedule should have the following skills: 

1. Automatic update of the group table 

After each entry of a match result the group table is automatically updated. 

All FIFA rules are taken into account, including fair play and drawing lots 

2. Automatic assignment of teams in the knockout phase 

 The FIFA regulations for the knockout phase will be incorporated. 

3. Multilingualism 

The team names and all small and large headings are displayed in English, Spanish, Italian, French and German. 

In addition, translations in your native language can be added. 

4. Choice of time zone 

 All time zones of the earth can be selected in a selection field, so that the date and time of the matches are dis-

played in your own local time. 

5. Colored highlighting of the current day's matches 

 The date of the matches that take place today is highlighted in color. 

6. Colored highlighting of missing entries 

 If matches have already taken place in the past few days and no results have been entered, the input fields are 

highlighted in color. 

7. Notice if an invalid tie was entered in the knockout phase 

 The matches in the knock-out phase will be decided by a penalty shoot-out at the latest. If a draw is entered for 

both the match result and the penalty shoot-out, a corresponding note will appear below the entry fields. 

8. Display of all cases of direct comparison in the group stage 

 If at the end of the group stage two or three teams are tied in points, goal difference and goals scored, a direct 

comparison takes place in which only the matches of these teams against each other are taken into account. 

These evaluations are shown on a separate worksheet. 

9. Easier adaptation to future world championships 

 The formulas are kept flexible enough to save a lot of work when adapting them for the next World Cup. At the 

next World Cup in 2026 in USA/Canada/Mexico there will be 16 groups of three instead of 8 groups of four and 

an additional sixteenth final. 

 

 

The programming language VBA (Visual Basic for Applications) is not used, because then the schedule can only be 

called up with Microsoft Excel. In addition, the majority of Excel fans are probably more familiar with formulas than 

with VBA. 

The so-called matrix formulas, which are difficult to understand, are avoided. Only in the last (optional) chapter 10 

there are used two matrix formulas, but they are easy to understand and are explained in detail. 

1 Structure of the schedule 

In the first step we create the template for the schedule. All fields remain empty at first because they later contain 

a formula that inserts the language-dependent content depending on the language selected. We call this first 

worksheet “World Cup”. It looks like this: 

The entries that can already be seen here are for illustration purposes only. They will be replaced by formulas later. 

The columns from column Z and the rows from row 98 are hidden. 

 

If you want to skip this step, you will find the finished result in the file named „WCup_Tut_1.xlsx“. 

2 Choice of language 

Before entries can be made anywhere, we need a worksheet with the translations of all team names and all labels 

into the various national languages. We are therefore creating a new worksheet with the name "Language". The 

illustration on the next two pages shows what this worksheet looks like. 

It contains the first 80 nations in the FIFA World Ranking (as of March 2022). 

For a particular group (e.g. Group A), we later select four nations by entering four of the consecutive numbers in 

the group table, e.g. 51 for Qatar, 1 for Brazil, 16 for Croatia and 30 for Nigeria (see chapter 4.1). 

This gives us maximum flexibility for future schedules as well. 

In the “my language” column, users can enter all translations in their own native language and then have a schedu-

le entirely in their native language. 

If you want to save yourself the work of creating this worksheet, you will find the finished result in the file named 

„WCup_Tut_2.xlsx“. 

 

The following describes how this Excel sheet is created and how it works. 

First, all texts must be entered in columns B, D, F, G, H, I, J and K. 

Column D will play an important role - see chapter 4. 

A selection field is setup in cell N1:  Menu  „Data  Data Tools  Data Validation“. 

We choose the option „List“. We specify the range  $B$3:$B$9  as the source for the list elements. In this range we 

enter the languages: „english, spanish, italian, ...“. 

Now we start with the first formulas. In cell C3 we enter the formula  =IF($N$1=B3;1;0). This means: if the language 

"english" was selected in cell N1, a „1“ appears in C3, otherwise a „0“. We now copy this formula down to cell C9 

(move the mouse to the lower right corner until the mouse pointer turns into a small black cross, then drag it down 

with the left mouse button pressed). 

A „1“ now appears to the right of the language that was selected in field N1. Unfortunately, however, someone can 

also click cell N1 and then press the "Delete" key. Then no language is selected at all. In this case, the language 

"english" should be selected (default value). 

To make that happen, let's change the formula in cell C3:  

=IF(AND($M$1<>B4,$M$1<>B5,$M$1<>B6,$M$1<>B7,$M$1<>B8,$M$1<>B9);1;0) 

That means:  If none of the other languages is selected, then there should be a „1“ for "english". 

In column E we now enter formulas which copy the texts of the selected language into column E. 

We start with cell E5: 

=IF($C$3,F5,IF($C$4,G5,IF($C$5,H5,IF($C$6,I5,IF($C$7,J5,IF($C$7,K5,IF(L5<>"",L5,F5))))))) 

The formula starts with  =IF($C$3,F5,...   It should actually be there  =IF($C$3=1,F5,... 

If there is a „1“ in C3, the English text from F5 should be copied into column E. Since Excel regards a „1“ as TRUE 

and a „0“ as FALSE, it is sufficient to write IF($C$3, here. 

If there is a „1“ in C3 (C3 is TRUE), the text from F5 is adopted in E5, otherwise there is another IF condition: 

IF($C$4,G5,...   If C4 is 1, the Spanish text is copied from cell G5 to E5, and so on. 

We copy the formula in E5 down to cell E166. In the lines where no translations take place, we delete the formulas, 

e.g. in E95, E96, E114, E115, ... The dollar signs in the formula are important: $C$3 should remain unchanged in the 

next line, but F5 should become F6, F7, etc. 

The auxiliary columns B, C and E can be hidden later. 

 

As a first test, we connect the cells E1 to U1 on the worksheet "World Cup" and enter the formula   

=Language!$E$115 . The title of our match schedule now appears in the selected language. 

 

The current status of the schedule can be found in the file named „WCup_Tut_2.xlsx“. 

3 Choice of time zone 

We will create a new worksheet called “TimeZone”. 

We enter the formula  =Language!$E$118  in cell A1 and merge cells A1 to F1. 

In column C we enter the currently existing time zones. We start with cell C4. In cell H1 we create a selection field 

that contains exactly this list of time zones (range $C$4:$C$39). The additional designations in column D are for 

convenience only and have no further functionality. 

The numbers in the range $B$4:$B$39 are decisive for the calculations of date and time. These are the shifts in 

hours compared to the coordinated universal time UTC. Central European Time CET, for example, is UTC+1. These 

numbers are not generated by a formula, but entered manually. 

The kick-off times for the individual matches are always given by FIFA in local time. So that this schedule can also be 

easily adapted for future world championships, we connect cells L5 and L6 and create the same selection field here 

as in H1. The time zone of the host country should be entered here. The columns from column K onwards will be 

hidden later. 

The time in Qatar is UTC+3. Therefore we select the time zone UTC+3 in this second selection field. 

We will take this time zone into account in the conversion formulas for the various local times. In this way, you only 

need to enter the official kick-off times of the host country for each World Cup and select the time zone of the host 

country in the second selection field. Everything else is then calculated automatically (see Section 4.2). 

In column B, the hour shifts are in a form that we can read. Excel saves the hours in fractions of a day in its date 

and time formats. So three hours would be 3/24 = 0.125. 

In Excel, a complete date with time (a timestamp) is nothing more than a decimal number indicating the number of 

days that have passed since 01.01.1900, 00:00 hours. The whole days are in front of the comma, the digits after the 

comma indicate the fraction of a day. So these digits contain the hours, minutes and seconds. 

If we add the decimal number 0.125 to a full Excel date, we have added three hours to it. That's exactly what we 

want to achieve when converting to another time zone. 

In cell E4 we enter the formula  =$B4/24  and copy it down to E39. So we have the corresponding decimal number 

for each time zone, which we have to add to a UTC time. 

The cells G4:I6 are merged and we enter the formula  =Language!$E$165. Cell G1 contains the formula =Language!

$E$161 and the merged cells K5:K6 contain the formula =Language!$E$162. 

We select the cells C4 and D4 and insert a conditional formatting with the formula  =$H$1=$C4. The selected time 

zone should be coloured green. The formatting is transferred to all time zones in columns C and D with the help of 

the "Format painter". 

Columns B and E will be hidden later. 

 

The current status of the schedule can be found in the file named „WCup_Tut_3.xlsx“. 

4 Grouping and entry of the matches 

The simplest way would be to enter the matches in the individual groups directly on the "World Cup" spreadsheet. 

In order to achieve maximum flexibility and make it as convenient as possible to adapt the match schedule for futu-

re tournaments, we will include two intermediate steps. 

In the first step, we make the grouping on an extra spreadsheet called "Groups". In the second step, we create an 

overview of the 64 matches of the tournament on a spreadsheet called "Matches". The transfer to the spreadsheet 

"World Cup" is then done automatically with the help of formulas. 

 

4.1 Grouping 

We create a new spreadsheet called "Groups". 

The names of the nations are already available in different languages on the "Language" spreadsheet. The grouping 

should now take place on the worksheet "Groups". 

In column B we enter the abbreviations A1, A2, ... H4. In column C 

we enter the numbers from the FIFA World Ranking List on the 

spreadsheet "Language", i.e. 51 for Qatar, 46 for Ecuador, 20 for 

Senegal, etc. 

For a new grouping, simply change these numbers. 

In order for the corresponding names to appear automatically in 

column D, we enter the following formula in cell D7: 

=VLOOKUP($C7,Language!$D$5:$E$94,2,0) 

The VLOOKUP function has four parameters. 

1st parameter:   $C7 

The content of C7 is searched for, in our example for "51". 

2nd parameter:   Language!$D$5:$E$94 

This is the range where searches are made and found. The search 

column is always the first column on the far left. 

3rd Parameter:   2 

The result is to be found in the second column. 

4th Parameter:   0 

The "0" (or also "FALSE") has the effect that when searching in the 

search column, only exact matches are found. 

So the formula looks in our FIFA World Ranking List to see which 

name belongs to number 51 and enters this name in cell D7. 

We copy the formula down to cell D45. 

4.2 Matches 

We'll create a new worksheet called "Matches" (see illustration on next page). 

The idea is this: 

We create an overview of all 64 matches on the "Matches" sheet. We number the matches from 1 to 64. These num-

bers are entered in column B. The order is determined by FIFA's actual scheme for running the tournament: first the 

group matches A1-A2, A3-A4, B1-B2, B3-B4 ect., then the knock-out matches. 

On the “World Cup” worksheet, our actual schedule, there is a box for each match 

with the venue, date, teams and result. Each of these boxes is given one of these 

numbers at the top left, for example the number 28. The formulas in the box then 

cause the data of match 28 to be displayed in the box. 

When adapting to a new World Cup, we are now free to decide where to place 

match No. 28 on the new schedule. 

 

In columns C and D of the spreadsheet "Matches", we first transfer the teams according to the usual FIFA scheme.  

Match no. 28 would be  E1-E3, i.e. team 1 of group E against team 3 of group E. 

In the columns I and J, we do not enter the names directly, but a formula that provides us with the names automati-

cally. Cell C4 contains A1. So we have to look up the name of the first team for match no. 1 on the sheet "Groups" 

under A1. We‘ll find this name in cell Groups!D7.  

The VLOOKUP function is available in Excel for such a lookup. Therefore we enter the following formula in cell I4 on 

the "Matches" worksheet:  =VLOOKUP(C4,Groups!$B$7:$D$45,3,0) 

It is important that the first parameter does not contain any dollar signs. So we can copy this formula to the right 

and down at the same time. We go to the lower right corner of cell I4 until the cursor turns into a small black cross 

and drag the formula over the entire range I4: J51. Now the team names appear in this range in the currently 

selected language. 

 

4.3 Kick-off times 

In column E we enter the date and time for each match and format the cells so that the date and time are displayed. 

In column F, the times should now be converted into the time zone that was selected in the cell TimeZone!H1 . 

We enter the following formula in F4: 

=E4-VLOOKUP(TimeZone!$L$5,TimeZone!$C$4:$E$39,3,0) 

+VLOOKUP(TimeZone!$H$1,TimeZone!$C$4:$E$39,3,0) 

The first VLOOKUP function looks up the table of all time zones to see which decimal number (positive or negative) 

corresponds to the local time of the host country. In the Qatar example, this is the number 3/24 = 0.125. 

This number (0.125) is first subtracted from the time at the venue (E4). Three hours are subtracted. That means it is 

converted to UTC. The expression VLOOKUP(TimeZone!$H$1,TimeZone!$C$4:$E$39,3,0) is now used to look up in 

the table of all time zones which decimal number (positive or negative) still has to be added to get the desired local 

time. 

We copy the formula in F4 down to F72. 

In cells F52, F61, F66, F69 and F71 we delete the formula again. 

4.4 Venues 

It is most convenient if we only have to enter the number of the venue for each match. We look up these numbers 

in the spreadsheet "Language" in column D and enter them in column G on the spreadsheet "Matches". 

We now enter the formula  =IFNA(VLOOKUP($G4,Language!$D$97:$E$112,2,0),"")  into cell H4 and copy it down 

to cell H72. In lines 52, 61, 66, 69 and 71 we delete the formula again. The names of the venues now appear in 

column H. 

The IFNA function has two parameters: 

1st parameter: the value that should appear in the cell 

2nd parameter: what is to appear in the cell in the event of the error "#N/A" (no value available). 

In our case, the cell remains empty if the VLOOKUP function finds nothing. 

Columns B, F, H, I and J  should be locked to prevent accidental overwriting. They are no longer changed. 

Columns C, D, E and G, on the other hand, will be readjusted at the next World Cup. 

 

The current status of the schedule can be found in the file named “WCup_Tut_4.xlsx”. 

5 Matches of the group stage 

On the „World Cup“ sheet we will enter the required formulas for the first group match and then copy these for-

mulas to all other matches in the group stage. The formulas for the Round of 16, quarter-finals, semi-finals and fi-

nals will be discussed and entered later (see chapter 8). 

With the help of the conditional formatting, we will highlight in color the date fields that contain today's date. So 

immediately you can see which matches are taking place today. 

Also with the help of the conditional formatting, we will highlight empty input fields of matches that are in the past. 

So immediately you can see which match results are still to be entered. 

 

5.1 Formulas for the matches 

We go to the "World Cup" sheet and fill in the following formulas: 

Cell B9: =Language!$E$124&" A" 

Cell E9: =Language!$E$124&" B" 

etc. until group H. This means that these headings always appear in the selected national language. 

FIFA has numbered all the matches from 1 to 64. We now enter these numbers according to the following scheme 

at the top left of the box with the match data: 

The first match in group A is match no. 1. So we enter a „1“ in cell A11. We enter a „2“ in cell A16. 

In each group, the matches are arranged in chronological order.  

Round of 16: 49, 50, 53, 54, 51, 52, 55, 56  (explanation of the sequence in chapter 8) 

Quarter-finals: 57, 58, 59, 60 

Semi-finals: 61, 62 

Third place: 63 

Finale: 64 

So there is a 49 in cell A48, a 50 in cell D48, a 53 in cell G48 etc. 

Now we are going to fill the boxes for the matches of the group stage with content. Normally, we would look on 

the "Matches" sheet to see which data belong to match no. 1 and transfer this data to the box. With the help of the 

VLOOKUP function, we let this transfer of data take place automatically. 

Group A Group B Group C Group D Group E Group F Group G Group H 

1 3 8 6 11 12 13 14 

2 4 7 5 10 9 16 15 

18 17 22 21 25 26 29 30 

19 20 24 23 28 27 31 32 

35 33 39 37 43 41 47 45 

36 34 40 38 44 42 48 46 

In cell B11 we enter the formula =VLOOKUP(A11;Matches!$B$4:$K$72;7;0). The search criterion in this formula is 

A11. In A11 there is a 1. 

This 1 is searched for on the "Matches" worksheet in the first column of the range B$4:$K$72. After it has been 

found in row 4 of the worksheet, the content in column 7 ("Al Bayt") is returned by the VLOOKUP function. 

“Al Bayt” now appears as the desired venue in cell B11. 

All other formulas have the same structure: 

Cell B12: =VLOOKUP(A11,Matches!$B$4:$K$72,5,0) 

Cell B13: =VLOOKUP(A11,Matches!$B$4:$K$72,8,0) 

Cell C13: =VLOOKUP(A11,Matches!$B$4:$K$72,9,0) 

A11 is a relative reference (without the dollar sign). That is why we can select range B11:C13 on the “World Cup” 

worksheet and copy it into the boxes for the other group matches. The data of the match with the corresponding 

number appear in each box. We try it out: For example, in cell M21 we change the number from 25 to 31. Imme-

diately we see match number 31 in this box. 

This flexibility also makes it easier for us to arrange the matches for the Round of 16 (see Chapter 8). 

 

5.2 Conditional formatting 

We click on the merged cells B12:C12 and select “Start  Conditional Formatting  New Rule” in the menu. We 

select the last option as the rule type and enter the formula =INT(B12)=TODAY(). 

We click on "Format ..." and choose a yellow background. We confirm all 

settings with OK. 

With “Home  Format Painter” we copy this conditional formatting to all 

matches of the group stage and the Round of 16. 

(Double click on "Format Painter" and click on all date fields) 

We click on the connected cells C57:E57 and give these cells the same 

type of conditional formatting, but here we enter the formula 

=INT(C57)=TODAY(). Then we transfer this conditional formatting to all 

date fields of the remaining knockout matches (quarter-finals, semi-

finals, finals). 

 

The conditional formatting of the input fields for the match results requires four formulas: 

Cell B14: =AND(B12<TODAY(),B14="") 

The conditional formatting is transferred to all input fields on the left for the group stage and the Round of 16, 

i.e. E14, H14, etc. 

Cell C14: =AND(B12<TODAY(),C14="") 

The conditional formatting is transferred to all input fields on the right for the group stage and the Round of 16, 

i.e. F14, I14, etc. 

Cell C59: =AND(C57<TODAY(),C59="") 

Transfer to all input fields on the left for the remaining KO matches 

Cell E59: =AND(C57<TODAY(),E59="") 

Transfer to all input fields on the right for the remaining KO matches  

With the help of the data validation we prevent nonsensical entries for the match results. 

We mark all input fields and select in the menu  Data  Data Tools  Data validation. 

We only allow whole numbers between 0 and 99.  

 

The current status of the schedule can be found in the file named “WCup_Tut_5.xlsx”. 

6 Fair play and lot 

In the event that two teams are tied at the end of the group stage in terms of the number of points, the goal diffe-

rence and the number of goals scored and their direct encounter also ended in a draw, the fair play score or the lot 

will decide on the better placement. At the 2018 World Cup, Japan and Senegal were affected. The fair play rating 

decided in favor of Japan and Senegal was eliminated. 

In this case, it is sufficient if you can enter a bonus point on the beneficiary team, which then brings this team to 

the better position. 

On a new worksheet with the name "Fair play and lot" we enable this entry of a bonus point. 

Cell B2: =Language!$E$117 

Cell B3: =Language!$E$160 

In the cells to the left of the headings - i.e. 

in A5, A11, A17, A23, D5, D11, D17 and 

D23 - we enter the letter of the relevant 

group and hide it using the custom for-

matting ";;;" (three semicolons). 

Cell B5: =Language!$E$124&" "&A5 

Cell C5: =Language!$E$152 

 

 

 

 

 

 

 

 

Cell B6: =VLOOKUP(A5&"1",Groups!$B$7:$D$45,3,0) 

Cell B7: =VLOOKUP(A5&"2",Groups!$B$7:$D$45,3,0) 

Cell B8: =VLOOKUP(A5&"3",Groups!$B$7:$D$45,3,0) 

Cell B9: =VLOOKUP(A5&"4",Groups!$B$7:$D$45,3,0) 

The last four formulas read the four team names from the "Groups" spreadsheet. 

We enter zeros in the cells C6 to C9. 

Now we select the range B5:C9 and copy it seven times to the positions B11, B17, B23, E5, E11, E17 and E23. 

 

The current status of the schedule can be found in the file named “WCup_Tut_6.xlsx”. 

7 Calculation of the group tables 

The aim is to display the current status of the group table in range B3:C6 on our schedule and to automatically en-

ter the order of the teams in the final table B42:B45. 

We create a new worksheet and call it "GrpA". It will look like this: 

We then create the other spreadsheets with the names "GrpB", "GrpC", etc. by making seven copies of the 

spreadsheet "GrpA" and only entering a "B" or a "C" etc. in cell P1! 

 

7.1 Transfer of the match results 

Consideration: 

We find the values for group A on the „World Cup“ sheet in columns B and C, those for group B in columns E and F, 

and so on. That is the only difference. 

Therefore, we first create a new spreadsheet with the name "References" and create a small table here in the ran-

ge B4:D12 in which we enter the two relevant columns for each group (see illustration). 

On the spreadsheet "GrpA" we enter the following formulas in cells B1 and D1: 

B1: =VLOOKUP($P$1,References!$B$5:$D$12,2,0) 

D1: =VLOOKUP($P$1,References!$B$5:$D$12,3,0) 

A "B" then appears in cell B1 and a "C" in cell D1. The letter in cell P1 thus determines which two column names are 

in cells B1 and D1. 

The formula in cell B4 could simply be:  =‘World Cup‘!B13.  But then this formula would only be correct for group A. 

Therefore, we will assemble this formula text with the help of the function INDIRECT. 

If we write  =INDIRECT(A1), in cell A1 the INDIRECT function expects a cell reference such as "Z8", and the function 

would return the cell contents of Z8. 

But if we write  =INDIRECT(„A1“), the INDIRECT function returns the cell contents of A1. 

The formula is identical to  =A1. 

We use this second case for our purposes. In the end, the formula should be identical to   

=INDIRECT(“‘World Cup‘!B13“). 

We will now assemble the character string within the brackets from individual parts. In Excel formulas, strings are 

concatenated with "&". 

It looks like this: “‘World Cup‘!“ & $B$1 & “13“  (without the spaces). There is a "B" in  $B$1. 

For the second match in group A, this string would have to be  “‘World Cup‘!“ & $B$1 & “18“, because the second 

match can be found in line 18. 

To be even more flexible, we create a small table of these line numbers in range  

References!G5:G10 and a small table of line numbers for the match results in range  

References!H5:H10  (see figure on the left). 

In our string we replace the "13" at the end with „References!$G5“: 

“‘World Cup‘!“ & $B$1 & References!$G5 

Our formula is now:  =INDIRECT(“‘World Cup‘!“&$B$1&References!$G5) . 

The advantage: We can copy this formula by dragging it down with the mouse, which automatically creates $G6, 

$G7, ... at the end and the line numbers are taken from our small table. 

Small disadvantage: If we change the table name "World Cup", the table name is not automatically updated in the-

se formulas because it is within a character string and is just ordinary text for Excel. So that we do not have to 

change all 24 formulas in the “GrpA” worksheet in such a case, we store the table name in a central location where 

we only have to change it once. 

On the worksheet "References", we enter the name "World Cup" in cell A1. Our formula must now get this name 

from this cell. The formula then looks like this: 

=INDIRECT("'"&References!$A$1&"'!"&$B$1&References!$G5) .         That is the final formula. 

The other three formulas are structured in the same way: 

Cell D4: =INDIRECT("'"&References!$A$1&"'!"&$D$1&References!$G5) 

Cell E4: =INDIRECT("'"&References!$A$1&"'!"&$B$1&References!$H5)          [goals1] 

Cell G4: =INDIRECT("'"&References!$A$1&"'!"&$D$1&References!$H5)          [goals2] 

So that the number of goals is only taken into account in matches in which both fields for goals are filled in, we ha-

ve to expand the formulas in cells E4 and G4 a little. The final formula says: If the first field is empty or the second 

field is empty, nothing is entered here in cell E4, otherwise the result. 

Cell E4: =IF(OR(goals1=““;goals2=““);““;goals1) 

The terms  INDIRECT(…)  must then be used for "goals1" and "goals2" (see above). 

Final formulas for the goals: 

Cell E4: 

=IF(OR(INDIRECT("'"&References!$A$1&"'!"&$B$1&References!$H5)="",INDIRECT("'"&References!

$A$1&"'!"&$D$1&References!$H5)=""),"",INDIRECT("'"&References!$A$1&"'!"&$B$1&References!$H5)) 

Cell G4: 

=IF(OR(INDIRECT("'"&References!$A$1&"'!"&$B$1&References!$H5)="",INDIRECT("'"&References!

$A$1&"'!"&$D$1&References!$H5)=""),"",INDIRECT("'"&References!$A$1&"'!"&$D$1&References!$H5)) 

Now we are ready: We select cells B4:G4 and copy the formulas down to row 9. Done! - We have transferred the 

six matches in Group A and the match results to our “GrpA” worksheet. 

We can now try out the flexibility of these formulas. For example, if we enter the letter "B" in cell P1, we imme-

diately see the matches of group B with their results! 

 

7.2 Ranking list - part 1 

In cell K4 we enter the formula  =IF(AND($E4<>"",$G4<>""),1,0). Here it is checked whether the match result has 

been entered in full. The formula is copied down to cell K9. In cell K10 we form the sum of the range K4:K9. So we 

have the number of fully registered matches. 

Cell H4: =IF($K4>0,IF($E4>$G4,3,IF($E4=$G4,1,0)),0) 

Cell I4: =IF($K4>0,IF($E4<$G4,3,IF($E4=$G4,1,0)),0) 

The two formulas are copied down. 

In the range H4:H9  3 points appear if team 1 has won the match, 1 point in the case of a tie and 0 if the match was 

lost or no match result was entered. Correspondingly, the points scored by team 2 appear in range I4:I9. 

 

In the range N4:W7 we are now creating a table with the four teams, in which we calculate a rating number for 

each team that includes all FIFA criteria. This number in column T will then determine the order of the four teams. 

In column N is the consecutive number from 1 to 4. In column O are the four teams. 

In cell O4 we enter the formula =VLOOKUP($P$1&ROW($A1),Groups!$B$7:$D$45,3,0)  and copy it down to cell 

O7. 

The first parameter of the function VLOOKUP is the expression $P$1 & ROW($A1).  This expression resolves to "A1" 

because there is an "A" in $P$1 and ROW($A1) returns a "1". The function VLOOKUP looks on the spreadsheet 

'Groups' to see which country name can be found under "A1". It then looks in cell O5 for "A2" and so on. 

Cell P4 should contain the number of points that the first team has achieved. To do this, we add all the points in 

column H if the team concerned is in column B:  SUMIF($B$4:$B$9,O4,$H$4:$H$9) . 

In addition, all points in column I are added if the team in question is in column D:   

SUMIF($D$4:$D$9,O4,$I$4:$I$9) 

The formula in P4 is therefore: 

=SUMIF($B$4:$B$9,O4,$H$4:$H$9)+SUMIF($D$4:$D$9,O4,$I$4:$I$9) 

The formula can be copied down to P7. 

In cell Q4 we enter the formula for the goal difference:  =R4-S4 . This can also be copied downwards. 

For the goals scored in column R and the goals conceded in column S, formulas with the same structure as for the 

number of points are used. 

Cell R4: =SUMIF($B$4:$B$9,$O4,$E$4:$E$9)+SUMIF($D$4:$D$9,$O4,$G$4:$G$9) 

Cele S4: =SUMIF($B$4:$B$9,$O4,$G$4:$G$9)+SUMIF($D$4:$D$9,$O4,$E$4:$E$9) 

In column U the direct comparison of three teams should be taken into account and in column V the direct compa-

rison of two teams. The formulas for this need preparation and will be discussed later (see Chapters 7.5 and 7.6). 

Column W should contain the bonus points from the “Fair play and lot” worksheet. The formula in cell W4 is: 

=SUMIFS(FairPlayPoints1;FairPlayTeams1;O4)+SUMIFS(FairPlayPoints2;FairPlayTeams2;O4) 

We define the four variables FairPlayPoints1, FairPlayTeams1, FairPlayPoints2 and FairPlayTeams2 in the Name Ma-

nager of Excel (menu Formulas  Name Manager) as follows: 

FairPlayPoints1: 'Fair play and lot'!$C$6:$C$27 

FairPlayTeams1: 'Fair play and lot'!$B$6:$B$27 

FairPlayPoints2: 'Fair play and lot'!$F$6:$F$27 

FairPlayTeams2: 'Fair play and lot'!$E$6:$E$27 

In columns C and F on the worksheet “Fair play and lot”, all numbers are added up if the respective team name is 

to the left. 

 

7.3 The Factors sheet 

The sorting of the four teams into the correct ranking is done according to the following principle: In column T of 

the “GrpA” worksheet there is a “ranking number” for each team, which decides on the rank. The team with the 

largest number comes first. 

This number is created in the following way: 

The number of points achieved in column P has the highest priority. This number of points is therefore multiplied 

by a very large factor. We take the number 10000. The goal difference in column Q has the second highest priority. 

It is multiplied by the lower factor 100 and added. The goals scored in column R have the third highest priority. 

They are multiplied by a factor of 1 and also added. The remaining criteria (direct comparisons and fair play or dra-

wing lots) have even smaller factors: 0.1, 0.01 and 0.0001. 

We store these factors on a separate worksheet called "Factors". There are no formulas here, just texts and num-

bers. It looks like this: 

The last factor will be explained later. 

The individual factors in cells B4 to B10 are given the following variable names in the Name Manager: 

B4: FactorPts (factor points) 

B5: FactorGD (factor goal difference) 

B6: FactorFor (factor goals scored) 

B7: FactorDirC3 (factor direct comparison of 3 teams) 

B8: FactorDirC2 (factor direct comparison of 2 teams) 

B9: FactorFairPlay (factor fairplay) 

B10: FactorRow (factor row) 

Thus, instead of  Factors!$B$4 , we can simply write FactorPts in the formulas. The formulas will then be easier to 

read. 

 

7.4 Ranking list - part 2 

The formulas in U4 and V4 will be discussed later (see Chapters 7.5 and 7.6). 

In cell T4 we enter the following formula:  

=$P4*FactorPts+(GDzero+$Q4)*FactorGD+$R4*FactorFor 

  +$U4*FactorDirC3+$V4*FactorDirC2+$W4*FactorFairPlay 

GDzero is added to the goal difference $Q4 so that this value cannot become negative. This formula is still incom-

plete and will be expanded. 

This gives us a ranking number that takes all FIFA criteria into account with the right priority. Using this ranking 

number, we can now sort the group table. The team with the highest ranking should come first. In the range 

N11:T14 we will generate a sorted list of the four teams. 

We enter the following formula in cell T11: 

=LARGE($T$4:$T$7,ROW(A1)) 

and copy it down to T14. 

The formula  =LARGE(Bereich;1)  returns the largest value of range, 

the formula  =LARGE(Bereich;2)  the second largest, 

the formula  =LARGE(Bereich;3)  the third largest, 

the formula  =LARGE(Bereich;4)  the fourth largest. 

In order to be able to simply copy the first formula in T14 downwards, the numbers 1, 2, 3 and 4 are generated au-

tomatically. The expression ROW(A1) returns the row number of A1, ie 1. 

When copying down, it becomes ROW(A2), ROW(A3), ROW(A4), so the numbers 2, 3 and 4 are generated. 

In range N11:T14 we now want to create the same group table as in range N4:T7, only this time sorted. To do this, 

the first step is to search for the ranking number of cell T11 in range T4:T7. For example, if it is found in cell T5, all 

values to the left of cell T5 (the values of cells O5:S5) are copied down to cells O11:S11. 

The VLOOKUP function is not suitable here because it can only transfer values to the right of the search column. In 

this case, a combination of the INDEX and MATCH functions helps. 

First we enter the current number in cells N11 to N14. In cell O11 we enter the following formula: 

=INDEX(O$4:O$7,MATCH($T11,$T$4:$T$7,0)) 

The dollar sign in front of the "O" is missing because the formula is to be copied to the right. Then "O" becomes "P" 

and so on. 

Explanation of the formula: 

The formula  =INDEX(O$4:O$7,3)  would return the third value from the range O$4:O$7, i.e. the content of cell O6. 

The "3" is now replaced by the expression MATCH($T11,$T$4:$T$7,0). The MATCH function returns the correct 

number because it searches for the content of $T11 in the range $T$4:$T$7 and returns the relative position of the 

value found. The third parameter of the MATCH function is a zero. This means that the function is looking for an 

exact match. 

The formula in O11 is copied to the right and down to S14 and the sorted group table is ready! 

However, the sorted group table still has one error: 

For example, if the second and third team have the same ranking, the MATCH function will only find the position of 

the second team, but no longer the position of the third team. The table then shows the name of the second team 

twice with their values, the third team is missing. So we have to avoid having two or more ranks equal. For examp-

le, if all four are the same because no match results have yet been entered, the first team appears four times in the 

sorted list. Instead, the four teams should appear in their original order. We achieve this by adding the row number 

multiplied by the factor FactorRow (= 0.000001) to the ranking number. 

The final formula in cell T4 is then: 

=$P4*FactorPts+(GDzero+$Q4)*FactorGD+$R4*FactorFor+$U4*FactorDirC3 

  +$V4*FactorDirC2+$W4*FactorFairPlay+(8-ROW())*FactorRow 

The ROW() function returns the row number of the cell in which the formula is located. The expression 8-ROW() 

generates the number 4 for T4, the number 3 for T5, the number 2 for T6 and the number 1 for T7. Thus, if all 

teams are equal, the team in line 4 appears at the top, the team in line 5 appears in the second position, and so on. 

 

7.5 Direct comparison of three teams 

In contrast to the UEFA European Championship, a direct comparison of three teams is very unlikely at the FIFA 

World Cup. This is because this case occurs at the European Championship when three teams are tied on points. At 

the FIFA World Cup, on the other hand, a direct comparison of three teams is only carried out if all three teams ha-

ve the same number of points, the same goal difference and the same number of goals scored. This is very unlikely, 

but not impossible, so that correctly this case must also be considered here. 

In order to carry out a direct comparison of three teams, we create four group tables with three teams each, in 

which only the matches are taken into account that these three teams have played against each other. 

For the team names we enter the following formulas: 

Cell O17: =$O$4 

Cell O18: =$O$5 

Cell O19: =$O$6 

Cell O20: =$O$7 

In cell P17 we enter the following formula: 

=SUMIFS($H$4:$H$9,$B$4:$B$9,$O17,$D$4:$D$9,"<>"&$O$20) 

+SUMIFS($I$4:$I$9,$B$4:$B$9,"<>"&$O$20,$D$4:$D$9,$O17) 

Parameters of the function SUMIFS: 

$H$4:$H$9 Sum range: the range in which the numbers are added under certain conditions 

$B$4:$B$9 Criteria range 1: the range that is checked for criterion 1 

$O17 Criterion 1: the content of $O17 is checked for agreement in criteria range 1 

$D$4:$D$9 Criteria range 2: the range that is checked for criterion 2 

"<>"&$O$20 Criterion 2: in Criteria range 2 we check whether the content of $O17 is not the same as the 

 content of $O$20 

The second function SUMIFS in the formula has the same structure. The points scored for each of the three teams 

are added up if they were not scored against the fourth team in cell O20. 

The formulas in R17 and S17 work on the same principle: 

Cell R17: 

=SUMIFS($E$4:$E$9,$B$4:$B$9,$O17,$D$4:$D$9,"<>"&$O$20) 

+SUMIFS($G$4:$G$9,$B$4:$B$9,"<>"&$O$20,$D$4:$D$9,$O17) 

Cell S17: 

=SUMIFS($G$4:$G$9,$B$4:$B$9,$O17,$D$4:$D$9,"<>"&$O$20) 

+SUMIFS($E$4:$E$9,$B$4:$B$9,"<>"&$O$20,$D$4:$D$9,$O17) 

Cell Q17: =R17-S17 

Cell T17: =P17*FactorPts+Q17*FactorGD+R17*FactorFor 

In a direct comparison, the points achieved, the goal difference and the goals scored are compared with each other. 

In cell T17, the ranking number stands for the direct comparison. 

The formulas from P17 to T17 are now copied down to line 19. 

In column U we now generate a ranking number for each of the three teams. 

Cell U17: =RANK(T17,T$17:T$19,1) 

The RANK function determines the rank of the number in T17 within the range T$17:T$19 . The third parameter, 

the "1", means that the largest number gets the lowest rank, ie rank 3. 

A direct comparison of three teams will only take place if all three teams have the same number of points, the sa-

me goal difference and the same number of goals scored. To check this, we generate a ranking number in the range 

Y4:Y7 that only takes these three criteria into account. 

In cell Y4 we enter the formula  =$P4*FactorPts+$Q4*FactorGD+$R4*FactorFor  and copy it down to Y7. 

If the criteria for the direct comparison of these three teams apply, these ranks are copied into column V, otherwi-

se zeros are entered in column V. This checking and copying is done by the formula in V17:   

=IF(AND($Y$4=$Y$5,$Y$4=$Y$6,$Y$4<>$Y$7),U17,0) 

The rest of the formulas look very similar: 

Group 2 

Cell O22: =$O$4 

Cell O23: =$O$5 

Cell O24: =$O$7 

Cell O25: =$O$6 

Cell P22: 

=SUMIFS($H$4:$H$9,$B$4:$B$9,$O22,$D$4:$D$9,"<>"&$O$25) 

+SUMIFS($I$4:$I$9,$B$4:$B$9,"<>"&$O$25,$D$4:$D$9,$O22) 

Cell Q22: =R22-S22 

Cell R22: 

=SUMIFS($E$4:$E$9,$B$4:$B$9,$O22,$D$4:$D$9,"<>"&$O$25) 

+SUMIFS($G$4:$G$9,$B$4:$B$9,"<>"&$O$25,$D$4:$D$9,$O22) 

Cell S22: 

=SUMIFS($G$4:$G$9,$B$4:$B$9,$O22,$D$4:$D$9,"<>"&$O$25) 

+SUMIFS($E$4:$E$9,$B$4:$B$9,"<>"&$O$25,$D$4:$D$9,$O22) 

Cell T22: =P22*FactorPts+Q22*FactorGD+R22*FactorFor 

Cell U22: =RANK(T22,T$22:T$24,1) 

Cell V22: =IF(AND($Y$4=$Y$5,$Y$4=$Y$7,$Y$4<>$Y$6),U22,0) 

 

Group 3 

Cell O27: =$O$4 

Cell O28: =$O$6 

Cell O29: =$O$7 

Cell O30: =$O$5 

Cell P27: 

=SUMIFS($H$4:$H$9,$B$4:$B$9,$O27,$D$4:$D$9,"<>"&$O$30) 

+SUMIFS($I$4:$I$9,$B$4:$B$9,"<>"&$O$30,$D$4:$D$9,$O27) 

Cell Q27: =R27-S27 

Cell R27: 

=SUMIFS($E$4:$E$9,$B$4:$B$9,$O27,$D$4:$D$9,"<>"&$O$30) 

+SUMIFS($G$4:$G$9,$B$4:$B$9,"<>"&$O$30,$D$4:$D$9,$O27) 

Cell S27: 

=SUMIFS($G$4:$G$9,$B$4:$B$9,$O27,$D$4:$D$9,"<>"&$O$30) 

+SUMIFS($E$4:$E$9,$B$4:$B$9,"<>"&$O$30,$D$4:$D$9,$O27) 

Cell T27: =P27*FactorPts+Q27*FactorGD+R27*FactorFor 

Cell U27: =RANK.EQ(T27,T$27:T$29,1) 

Cell V27: =IF(AND($Y$4=$Y$6,$Y$4=$Y$7,$Y$4<>$Y$5),U27,0) 

Group 4 

Cell O32: =$O$5 

Cell O33: =$O$6 

Cell O34: =$O$7 

Cell O35: =$O$4 

Cell P32: 

=SUMIFS($H$4:$H$9,$B$4:$B$9,$O32,$D$4:$D$9,"<>"&$O$35) 

+SUMIFS($I$4:$I$9,$B$4:$B$9,"<>"&$O$35,$D$4:$D$9,$O32) 

Cell Q32: =R32-S32 

Cell R32: 

=SUMIFS($E$4:$E$9,$B$4:$B$9,$O32,$D$4:$D$9,"<>"&$O$35) 

+SUMIFS($G$4:$G$9,$B$4:$B$9,"<>"&$O$35,$D$4:$D$9,$O32) 

Cell S32: 

=SUMIFS($G$4:$G$9,$B$4:$B$9,$O32,$D$4:$D$9,"<>"&$O$35) 

+SUMIFS($E$4:$E$9,$B$4:$B$9,"<>"&$O$35,$D$4:$D$9,$O32) 

Cell T32: =P32*FactorPts+Q32*FactorGD+R32*FactorFor 

Cell U32: =RANK(T32,T$32:T$34,1) 

Cell V32: =IF(AND($Y$5=$Y$6,$Y$5=$Y$7,$Y$5<>$Y$4),U32,0) 

 

We enter the following formula in cell U4: 

=SUMIFS($V$17:$V$34,$O$17:$O$34,$O4) 

That means: 

In the range V17:V34  all numbers that belong to the first team ($O4) are added. So if a direct comparison is made 

for this team, a number greater than zero will appear in cell U4. This number is multiplied by the factor FactorDirC3 

and taken into account in the formula for the rank number (in cell T4) (see Chapter 7.4). 

We copy the formula in U4 down to U7. This concludes the direct comparison of three teams and we come to the 

direct comparison of two teams. 

7.6 Direct comparison of two teams 

In order to take into account the direct comparison of two teams, we first create a small table with the results of 

the direct encounters in range  D17:J21 . 

We transfer the team names to cells D18 to D21 with the formulas =$O$4, =$O$5, =$O$6 and =$O$7. The full na-

mes must be here! The first three letters of the team names are sufficient for the labeling in cells E17 to H17 above: 

Cell E17: =LEFT($O$4;3) 

Correspondingly in the other cells. 

The number of goals scored by team 1 against team 2 should be in cell F18. Cell E19 then shows the number of 

goals that team 2 scored against team 1. We enter the following formula in F18: 

=SUMIFS($E$4:$E$9,$B$4:$B$9,$O$4,$D$4:$D$9,$O$5) 

+SUMIFS($G$4:$G$9,$B$4:$B$9,$O$5,$D$4:$D$9,$O$4) 

In the range $E$4:$E$9, the numbers are added if team 1 is on the left and team 2 is on the right. In the range 

$G$4:$G$9, the numbers are added if team 2 is on the left and team 1 is on the right. This way we get exactly the 

number of goals that team 1 scored against team 2. 

The rest of the formulas work in the same way. They are: 

Cell G18: =SUMIFS($E$4:$E$9,$B$4:$B$9,$O$4,$D$4:$D$9,$O$6) 

 +SUMIFS($G$4:$G$9,$B$4:$B$9,$O$6,$D$4:$D$9,$O$4) 

Cell H18: =SUMIFS($E$4:$E$9,$B$4:$B$9,$O$4,$D$4:$D$9,$O$7) 

 +SUMIFS($G$4:$G$9,$B$4:$B$9,$O$7,$D$4:$D$9,$O$4) 

Cell E19: =SUMIFS($E$4:$E$9,$B$4:$B$9,$O$5,$D$4:$D$9,$O$4) 

 +SUMIFS($G$4:$G$9,$B$4:$B$9,$O$4,$D$4:$D$9,$O$5) 

Cell G19: =SUMIFS($E$4:$E$9,$B$4:$B$9,$O$5,$D$4:$D$9,$O$6) 

 +SUMIFS($G$4:$G$9,$B$4:$B$9,$O$6,$D$4:$D$9,$O$5) 

Cell H19: =SUMIFS($E$4:$E$9,$B$4:$B$9,$O$5,$D$4:$D$9,$O$7) 

 +SUMIFS($G$4:$G$9,$B$4:$B$9,$O$7,$D$4:$D$9,$O$5) 

Cell E20: =SUMIFS($E$4:$E$9,$B$4:$B$9,$O$6,$D$4:$D$9,$O$4) 

 +SUMIFS($G$4:$G$9,$B$4:$B$9,$O$4,$D$4:$D$9,$O$6) 

Cell F20: =SUMIFS($E$4:$E$9,$B$4:$B$9,$O$6,$D$4:$D$9,$O$5) 

 +SUMIFS($G$4:$G$9,$B$4:$B$9,$O$5,$D$4:$D$9,$O$6) 

Cell H20: =SUMIFS($E$4:$E$9,$B$4:$B$9,$O$6,$D$4:$D$9,$O$7) 

 +SUMIFS($G$4:$G$9,$B$4:$B$9,$O$7,$D$4:$D$9,$O$6) 

Cell E21: =SUMIFS($E$4:$E$9,$B$4:$B$9,$O$7,$D$4:$D$9,$O$4) 

 +SUMIFS($G$4:$G$9,$B$4:$B$9,$O$4,$D$4:$D$9,$O$7) 

Cell F21: =SUMIFS($E$4:$E$9,$B$4:$B$9,$O$7,$D$4:$D$9,$O$5) 

 +SUMIFS($G$4:$G$9,$B$4:$B$9,$O$5,$D$4:$D$9,$O$7) 

Cell G21: =SUMIFS($E$4:$E$9,$B$4:$B$9,$O$7,$D$4:$D$9,$O$6) 

 +SUMIFS($G$4:$G$9,$B$4:$B$9,$O$6,$D$4:$D$9,$O$7) 

A direct comparison of two teams will only take place if both teams have the same number of points, the same goal 

difference and the same number of goals scored. To check this, we use the ranking numbers in the range Y4:Y7 

which only consider these three criteria. 

In the range I18:I21, a team is now given a 3 (three points) if it has won a direct comparison. If there is a tie in the 

direct comparison, it receives a 1 (one point). Otherwise there are all zeros in these cells. 

To do this, we enter the following formula in cell I18: 

=IF(AND($Y$4=$Y$5,$Y$4<>$Y$6,$Y$4<>$Y$7),IF(F18>E19,3,IF(F18=E19,1,0)),0) 

+IF(AND($Y$4=$Y$6,$Y$4<>$Y$5,$Y$4<>$Y$7),IF(G18>E20,3,IF(G18=E20,1,0)),0) 

+IF(AND($Y$4=$Y$7,$Y$4<>$Y$5,$Y$4<>$Y$6),IF(H18>E21,3,IF(H18=E21,1,0)),0) 

The first IF function means: If team 1 is tied with team 2 in points, goal difference and goals scored ($Y$4=$Y$5), 

but otherwise with no other team ($Y$4<>$Y$6,$Y$4<>$Y$7), then she gets 3 points for a win against team 2 

(F18>E19), 1 point for a draw (F18=E19) and 0 points for a defeat. 

The second and third IF functions test whether a direct comparison of team 1 with team 3 or team 4 takes place. 

The formulas for cells I19, I20 and I21 are accordingly: 

Cell I19: 

=IF(AND($Y$5=$Y$4,$Y$5<>$Y$6,$Y$5<>$Y$7),IF(E19>F18,3,IF(E19=F18,1,0)),0) 

+IF(AND($Y$5=$Y$6,$Y$5<>$Y$4,$Y$5<>$Y$7),IF(G19>F20,3,IF(G19=F20,1,0)),0) 

+IF(AND($Y$5=$Y$7,$Y$5<>$Y$4,$Y$5<>$Y$6),IF(H19>F21,3,IF(H19=F21,1,0)),0) 

Cell I20: 

=IF(AND($Y$6=$Y$4,$Y$6<>$Y$5,$Y$6<>$Y$7),IF(E20>G18,3,IF(E20=G18,1,0)),0) 

+IF(AND($Y$6=$Y$5,$Y$6<>$Y$4,$Y$6<>$Y$7),IF(F20>G19,3,IF(F20=G19,1,0)),0) 

+IF(AND($Y$6=$Y$7,$Y$6<>$Y$4,$Y$6<>$Y$5),IF(H20>G21,3,IF(H20=G21,1,0)),0) 

Cell I21: 

=IF(AND($Y$7=$Y$4,$Y$7<>$Y$5,$Y$7<>$Y$6),IF(E21>H18,3,IF(E21=H18,1,0)),0) 

+IF(AND($Y$7=$Y$5,$Y$7<>$Y$4,$Y$7<>$Y$6),IF(F21>H19,3,IF(F21=H19,1,0)),0) 

+IF(AND($Y$7=$Y$6,$Y$7<>$Y$4,$Y$7<>$Y$5),IF(G21>H20,3,IF(G21=H20,1,0)),0) 

In cell V4 we enter the formula =$I18  and copy this formula down to V7. 

The number in cell V4 is multiplied by the factor FactorDirC2 and taken into account in the formula for the ranking 

number in cell T4 (see Chapter 7.4). 

The direct comparison of two teams is now done and we can now finish creating the group table. 

7.7 Ranking list - part 3 

We now make seven copies of the worksheet GrpA, which we call GrpB, GrpC, ..., GrpH. 

A copy of a worksheet can be created by right-clicking on the name of the worksheet at the bottom, then selecting 

"Move or copy ..." from the context menu. Don't forget to tick the "Create a copy" box! 

 

On each copy, we only need to change one cell: cell P1. 

For group B we enter a "B" here, for group C a "C" and so on. 

 

Our formulas in cells B1, D1 and in the ranges B4:G9 and O4:O7 have the effect that just by changing cell P1 the 

whole table is filled with the data of the correct group. 

7.8 Group table on the schedule 

On the worksheet "World Cup" we enter the formula =Language!$E$128 in the cells C2, F2, I2, L2, O2, R2, U2 and 

X2. The spaces in the text  “  Pts.     Goals“  ensure the correct position of the headings "Pts." And "Goals". 

Cell B3: =GrpA!O11 

This formula is copied down to cell B6. 

Cell E3: =GrpB!O11 

This formula is copied down to cell E6. 

We enter the corresponding formulas for the remaining six groups. 

To display the points and the goals, we enter the following formula in cell C3: 

 ="   "&GrpA!$P11&"        "&GrpA!$R11&Language!$E$164&GrpA!$S11. 

Here, too, the inserted spaces ensure correct positioning. 

In Language!$E$164 there is a separator for the goal difference (colon or minus) surrounded by a space. 

This formula is also copied down to cell C6. 

In cell F3 we enter the following formula: 

="   "&GrpB!$P11&"        "&GrpB!$R11&Language!$E$164&GrpB!$S11 

We enter the corresponding formulas for the remaining six groups. 

 

In the Final table in lines 41 to 45, the fields should remain empty if no match has been entered in the relevant 

group. 

We enter the following formulas in cells B41 and B42: 

Cell B41: =Language!$E$134 

Cell B42: =IF(GrpA!$K$10>0,GrpA!$O11,"") 

The formula in B42 is copied down to cell B45. 

The Final tables of the other groups have corresponding formulas. “GrpA” changes to “GrpB” etc. 

 

The current status of the schedule can be found in the file named “WCup_Tut_7.xlsx”. 

8 Knockout phase 

We have prepared all match data for the group stage to such an extent that we only need to enter the number of 

the match in the top left next to each match box on the “World Cup” worksheet. The match data is then automati-

cally taken from the "Matches" sheet. 

We will also apply this principle in the Round of 16. 

There will be an exception for the other games in the knockout phase (quarter-finals, semi-finals, finals): 

The team names for the individual encounters are not prepared on the "Matches" sheet. 

Actually, it would be best to use this principle for all games up to the final. But from the quarter-finals onwards, the 

team names depend on entries on the “World Cup” worksheet. We would need difficult-to-read formulas with unne-

cessary computational effort to first generate the team names on the "Matches" sheet and then transfer them to the 

"World Cup" sheet. 

 

8.1 Round of 16 

On the “Matches” worksheet in cells C53 and D53, we see that match number 49 is match 1A against 2B. The corres-

ponding team names should now appear in columns I and J - but not statically, but depending on the entries in co-

lumns C and D - just like in the group matches. 

To do this, we create a small auxiliary table in the range M53:N68. It has the following entries: 

The team name for 1A is in the cell 'World Cup'!$B$42, the team name for 2A is in the cell 'World Cup'!$B$43 and so 

on. 

In cell I53 we enter the formula =VLOOKUP(C53,$M$53:$N$68,2,0) and copy it down and to the right over the whole 

range I53:J60. The team names of the Round of 16 matches now appear. 

 M N 

53 1A ='World Cup'!$B$42 

54 2A ='World Cup'!$B$43 

55 1B ='World Cup'!$E$42 

56 2B ='World Cup'!$E$43 

57 1C ='World Cup'!$H$42 

58 2C ='World Cup'!$H$43 

59 1D ='World Cup'!$K$42 

60 2D ='World Cup'!$K$43 

61 1E ='World Cup'!$N$42 

62 2E ='World Cup'!$N$43 

63 1F ='World Cup'!$Q$42 

64 2F ='World Cup'!$Q$43 

65 1G ='World Cup'!$T$42 

66 2G ='World Cup'!$T$43 

67 1H ='World Cup'!$W$42 

68 2H ='World Cup'!$W$43 

The headings "Round of 16 - 1", "Round of 16 - 2", etc. are not officially given. We choose them according to the 

chronological order of the Round of 16 matches. However, the headings can just as well be numbered from left to 

right. 

To make it as easy as possible to adapt the chronological order to future World Cup tournaments, we enter the 

numbers from 1 to 8 in the desired order in cells K53 to K60. The rest will be done by formulas. 

At the 2022 World Cup in Qatar, Match No. 49 (Round of 16 Match 1A-2B) is first, it gets the number 1. 

Match No. 50 (Round of 16 Match 1C-2D) is second, it gets the number 2. 

Match No. 51 (Round of 16 Match 1B-2A) will be the fourth, it gets the number 4. 

In this way, the chronological order is 1, 2, 4, 3, 5, 6, 7, 8. 

 

Now we fill the match plan with the dates of the Round of 16. 

On the "World Cup" spreadsheet, we enter the following formula in cell B47: 

=VLOOKUP(VLOOKUP(A48,Matches!$B$53:$K$60,10,0),Language!$D$136:$E$143,2,0) 

The inner VLOOKUP function  VLOOKUP(A48,Matches!$B$53:$K$60,10,0)  takes the match number "49" in cell A48 

and then looks up the number for this match in column K on the spreadsheet 'Matches'. 

It finds the number "1". 

The outer VLOOKUP function now looks on the spreadsheet 'Language' in the range $D$136:$E$143  to see which 

heading belongs to the number "1", and finds there the language-dependent heading for the first Round of sixteen, 

in English for example "Round of 16 - 1". The idea of translating only the word "Round of 16" and adding a 1 can 

possibly lead to conflicts with languages where the 1 is not at the end. 

We copy the formulas for range B48:C50 from one of the group games: for example, select range B36:C38, copy 

with “Ctrl C”, then click on cell B48 and paste with “Ctrl V”. 

In cell B52 we enter:  =Language!$E$153  ("Penalty shoot-out:"). Cells B52 and C52 are merged. 

They get a conditional formatting (font colour red) with the formula =AND(B51=C51;B51<>"";C51<>""). 

In line 54 we enter the following two formulas: 

Cell B54: =VLOOKUP(A48,Matches!$B$4:$K$72,2,0) 

Cell C54: =VLOOKUP(A48,Matches!$B$4:$K$72,3,0) 

Now we select the range B47:C54 and copy it into the other Round of 16 games. 

 

Explanation of the order of the Round of 16 games on the match schedule: 

If we follow the matches backwards from the final, we can see in the FIFA plan that in the final, the winner of match 

61 will play against the winner of match 62. In match 61, the winner of match 57 plays against the winner of match 

58. In match 57, the winner of match 49 plays against the winner of match 50. This gives us the following order of 

the Round of 16 games: 49, 50, 53, 54, 51, 52, 55, 56. 

       64        Final 

   61        62    1/2-Final 

 57    58    59    60  1/4-Final 

49  50  53  54  51  52  55  56 1/8-Final 

8.2 Quarter-finals, semi-finals and finals 

For the quarter-finals we enter the numbers 2, 1, 4, 3 in the cells K62:K65 of the spreadsheet 'Matches'. They result 

from the chronological order of the quarter-finals. 

Since the order of the semi-finals and the finals is always the same, we enter the corresponding formulas in this 

case: 

Cell K67: =Language!$E$148 

Cell K68: =Language!$E$149 

Cell K70: =Language!$E$150 

Cell K72: =Language!$E$151 

On the 'World Cup' spreadsheet, we enter the following formulas for the first quarter-final match: 

Cell C57: =VLOOKUP(VLOOKUP(B58,Matches!$B$62:$K$65,10,0),Language!$D$144:$E$147,2,0) 

Cell C58: =VLOOKUP(B58,Matches!$B$4:$K$72,7,0) 

Cell C59: =VLOOKUP(B58,Matches!$B$4:$K$72,5,0) 

Cell C62: =Language!$E$153 

Cell C64: =VLOOKUP(B58,Matches!$B$62:$K$72,2,0) 

Cell E64: =VLOOKUP(B58,Matches!$B$62:$K$72,3,0) 

The cells  C62:E62  are merged and get a conditional formatting for red font color with the formula 

=AND(C61=E61;C61<>"";E61<>""). 

We select the range C57:E64 and copy it into the boxes of the remaining quarter-final matches and the first semi-

final match. The formula for the heading is no longer correct in the semi-final and the error "#N/A" (no value 

available) appears. From the semi-final onwards, we have constant headings with a different formula. In cell F67 we 

therefore enter: 

Cell F67: =VLOOKUP(E68,Matches!$B$62:$K$72,10,0) 

We select the range F67:H74 and copy it into the boxes of the remaining semis or finals. 

 

At this point we have the possibility of further optimization: 

On the 'World Cup' spreadsheet, the numbers of the matches in the KO round (round of 16, quarter-finals and 

semi-finals) can be generated automatically using formulas. This has the advantage that once the match plan has 

been filled out on the 'Matches' spreadsheet, all the matches in the KO round are automatically arranged correctly. 

On the 'World Cup' spreadsheet, cell C65 has the content "W49". This means that the first round of 16 match (far 

left) must be match number 49. We therefore replace the number 49 in cell A48 with the formula 

=RIGHT($C$64,2)*1. 

The last two characters ("49") are truncated from the text "W49". Multiplying by 1 converts the truncated text to 

the number 49. This conversion is necessary because the LOOKUP functions in the range B47:C54 would not find 

the text "49" in the Matches worksheet. They would return the error code #N/A. 

We proceed accordingly with the other match numbers of the KO round. For example, in cell D48 we enter the for-

mula =RIGHT($E$64,2)*1. Cell H58 then contains the formula =RIGHT($H$74,2)*1 and cell E68 contains the formu-

la =RIGHT($L$92,2)*1. 

Cell K76 contains the number 63 and cell K86 contains the number 64. 

Now only the team names are missing. 

In cell C60 we enter the following formula: 

=IF(AND(B51<>"",C51<>""),IF(B51>C51,B50,IF(B51<C51,C50,IF(AND(B53<>"",C53<>""),IF(B53>C53,B50,IF

(B53<C53,C50,"")),""))),"") 

The winner of match 49 is said to be here. The formula consists of six nested IF functions. The first IF function 

checks whether the match result is entered completely in cells B51 and C51. If not, no team name is entered. Ot-

herwise: The next two IF functions check whether one of the two teams has won. In this case, the winner is en-

tered. Otherwise (draw): Now it is checked whether the result of the penalty shoot-out is complete and no draw 

has been entered. Accordingly, the winner or nothing is entered for the team name. 

Here are the remaining formulas for the team names: 

Cell E60: =IF(AND(E51<>"",F51<>""),IF(E51>F51,E50,IF(E51<F51,F50,IF(AND(E53<>"",F53<>""),IF

(E53>F53,E50,IF(E53<F53,F50,"")),""))),"") 

Cell I60: =IF(AND(H51<>"",I51<>""),IF(H51>I51,H50,IF(H51<I51,I50,IF(AND(H53<>"",I53<>""),IF(H53>I53,H50,IF

(H53<I53,I50,"")),""))),"") 

Cell K60: =IF(AND(K51<>"",L51<>""),IF(K51>L51,K50,IF(K51<L51,L50,IF(AND(K53<>"",L53<>""),IF

(K53>L53,K50,IF(K53<L53,L50,"")),""))),"") 

Cell O60: =IF(AND(N51<>"",O51<>""),IF(N51>O51,N50,IF(N51<O51,O50,IF(AND(N53<>"",O53<>""),IF

(N53>O53,N50,IF(N53<O53,O50,"")),""))),"") 

Cell Q60: =IF(AND(Q51<>"",R51<>""),IF(Q51>R51,Q50,IF(Q51<R51,R50,IF(AND(Q53<>"",R53<>""),IF

(Q53>R53,Q50,IF(Q53<R53,R50,"")),""))),"") 

Cell U60: =IF(AND(T51<>"",U51<>""),IF(T51>U51,T50,IF(T51<U51,U50,IF(AND(T53<>"",U53<>""),IF

(T53>U53,T50,IF(T53<U53,U50,"")),""))),"") 

Cell W60: =IF(AND(W51<>"",X51<>""),IF(W51>X51,W50,IF(W51<X51,X50,IF(AND(W53<>"",X53<>""),IF

(W53>X53,W50,IF(W53<X53,X50,"")),""))),"") 

Cell F70: =IF(AND(C61<>"",E61<>""),IF(C61>E61,C60,IF(C61<E61,E60,IF(AND(C63<>"",E63<>""),IF

(C63>E63,C60,IF(C63<E63,E60,"")),""))),"") 

Cell H70: =IF(AND(I61<>"",K61<>""),IF(I61>K61,I60,IF(I61<K61,K60,IF(AND(I63<>"",K63<>""),IF

(I63>K63,I60,IF(I63<K63,K60,"")),""))),"") 

Cell R70: =IF(AND(O61<>"",Q61<>""),IF(O61>Q61,O60,IF(O61<Q61,Q60,IF(AND(O63<>"",Q63<>""),IF

(O63>Q63,O60,IF(O63<Q63,Q60,"")),""))),"") 

Cell T70: =IF(AND(U61<>"",W61<>""),IF(U61>W61,U60,IF(U61<W61,W60,IF(AND(U63<>"",W63<>""),IF

(U63>W63,U60,IF(U63<W63,W60,"")),""))),"") 

Cell L78: =IF(AND(F71<>"",H71<>""),IF(F71<H71,F70,IF(F71>H71,H70,IF(AND(F73<>"",H73<>""),IF

(F73<H73,F70,IF(F73>H73,H70,"")),""))),"") 

Cell N78: =IF(AND(R71<>"",T71<>""),IF(R71<T71,R70,IF(R71>T71,T70,IF(AND(R73<>"",T73<>""),IF

(R73<T73,R70,IF(R73>T73,T70,"")),""))),"") 

Cell L88: =IF(AND(F71<>"",H71<>""),IF(F71>H71,F70,IF(F71<H71,H70,IF(AND(F73<>"",H73<>""),IF

(F73>H73,F70,IF(F73<H73,H70,"")),""))),"") 

Cell N88: =IF(AND(R71<>"",T71<>""),IF(R71>T71,R70,IF(R71<T71,T70,IF(AND(R73<>"",T73<>""),IF

(R73>T73,R70,IF(R73<T73,T70,"")),""))),"") 

8.3 Notice when entering an invalid match result 

A draw is not possible in the knockout phase, as the match will definitely be decided by a penalty shoot-out. If a tie 

is entered as a match result, a note should appear under the match result. 

We merge cells B55 and C55 and enter the following formula: 

=IF(AND(B51<>"",C51<>"",B51=C51,B53<>"",C53<>"",B53=C53),Language!$E$159,"") 

We copy this formula into cells E55, H55, K55, N55, Q55, T55 and W55. 

The note appears when the match result and the penalty shoot-out result have been entered in full, but no winner 

emerges. 

We connect cells C65, D65 and E65 and fill in the following formula: 

=IF(AND(C61<>"",E61<>"",C61=E61,C63<>"",E63<>"",C63=E63),Language!$E$159,"") 

We copy this formula into cells I65, O65, U65, F75, R75, L83 and L93. 

 

Finally, the announcement of the new world champion: 

Cell R86: =Language!$E$135 

Cell R88: =IF(AND(L89<>"",N89<>""),IF(L89>N89,L88,IF(L89<N89,N88,IF(AND(L91<>"",N91<>""),IF

(L91>N91,L88,IF(L91<N91,N88,"")),""))),"") 

The ranges R86:W87 and R88:W89 are each merged to one cell. 

 

The match plan is now ready. The rest (chapters 9 and 10) is additional convenience. 

The current status of the schedule can be found in the file named “WCup_Tut_8.xlsx”. 

9 Optional:  The thing with the red dot 

If two or more teams cannot be distinguished at the end of the group stage due to their points, goals and direct 

comparisons, a decision will be made on the basis of the fair play rating or a drawing of lots. The user of this match 

plan must then enter additional bonus points on the “Fair play and lot” worksheet. 

So that the occurrence of such a case is clearly recognizable, the affected groups are marked with a red dot. The red 

dot disappears again when the order has been clarified by entering bonus points. 

To make this even more transparent for the user, we also set up an eye-catching red or green message, for example 

"Groups with fair play valuation or drawing lots: A, C and G." 

or 

"The placement has been clarified in all groups." 

 

9.1 The red dot 

In cell V11 on the „GrpA“ worksheet we enter the following formula: 

=IF(AND(OR(TRUNC($T$11,4)=TRUNC($T$12,4),TRUNC($T$11,4)=TRUNC($T$13,4),TRUNC($T$11,4)=TRUNC

($T$14,4),TRUNC($T$12,4)=TRUNC($T$13,4),TRUNC($T$12,4)=TRUNC($T$14,4),TRUNC($T$13,4)=TRUNC

($T$14,4)),$K$10>0),1,0) 

In detail: 

TRUNC($T$11,4) means that the rank number in T11 (e.g. 30102.000004) is reduced to four decimal places. This 

cuts off the line number and leaves the ranking number, which contains all FIFA criteria from the number of points 

to the drawing of lots. 

If we call these abbreviated ranks R1, R2, R3 and R4, then the formula says: If at least two of these rank numbers 

are the same (R1 = R2 or R1 = R3 or ... R3 = R4) and at least one match result is entered in this group ($K$10>0), 

then a 1 appears in cell V11, otherwise a 0. 

If there is a 1 in cell V11, a red dot should be displayed in the relevant group. 

On the “World Cup” worksheet, we merge cells B8 and C8. The red dot should appear here. 

Therefore we enter the formula  =IF(GrpA!$V$11>0,"•","")  here. 

The dot can first be created in any cell with “Insert  Symbol” and then inserted into the formula with copy and 

paste. 

The formula in GrpA! V11 is now copied into the "GrpB", "GrpC" etc. worksheets. 

On the “World Cup” worksheet, the formula in B8 is copied to E8, H8 etc. 

"GrpA" is changed to "GrpB", "GrpC" etc. 

9.2 The message 

We create a new worksheet with the name "Distinctness". It should look like this: 

In cells B4 to B11 we write the letters from A to H. 

In cells C4 to C11 we enter the formulas  =GrpA!$V$11 ,  =GrpB!$V$11 , … , =GrpH!$V$11. 

This gives us an overview of the groups in which the fair play valuation or the lot will decide. These are all groups 

for which there is a 1 in column C - in the figure above, groups B, D and E. 

In this case, a message should appear on the schedule with the following text: 

Groups with fair play valuation or drawing lots: B, D and E. 

The first part of the message (groups with fair play valuation or drawing lots: ) is a fixed text. The second part 

(B, D and E.) is variable text that looks different depending on the situation. For example, it can also be: D and H. 

To create this second part of the text, we first enter a formula in cell D4 for the number of groups with 

fair play valuation/drawing lots:  =COUNTIF($C$4:$C$11,">0"). 

In column E we enter the letter of the group if the group is not clear. 

Cell E4: =IF($C4>0,$B4,"") 

The formula is copied down to E11. 

We enter the following formulas in cells F4 and G4: 

Cell F4: =LEN($E4) 

Cell G4: =SUM($F$4:$F4) 

and copy both formulas down to line 11. It is important that there is no dollar sign in front of the last 4 in the ex-

pression $F$4:$F4. When copying, $F4 becomes $F5, etc. 

Side note: Column F is only an auxiliary column to avoid the use of the SUMPRODUCT function, which behaves like a matrix formula and is therefore more 

difficult to understand. You could also omit column F and enter the formula  = SUMPRODUCT(LEN($E$4:$E4))  in column G. 

The numbers in column G now tell us the position of the three ambiguous groups in the list of these groups. The B 

in line 5 is in the first position. At the 3 in cell G11 we can see that there are two more ambiguous groups. Therefo-

re, a comma must follow after the B. The D in line 7 is in the second position. At the 3 in cell G11 we see that this is 

the penultimate position and must be followed by an “and” after the D. The E in line 8 is in the third position. At the 

3 in cell G11 we can see that this is the last position. Nothing more has to follow the E. The final period is placed by 

the formula in cell J4. 

This logic is summarized in the formulas in column H. 

Cell H4: =IF($E4<>"",$E4&IF($G4=$G$11,"",IF($G4<$G$11-1,", ",Language!$E$158)),"") 

The formula is copied down. The expression Language!$E$158 is the word “and” in the selected language. 

In cell H13 we put all parts together into one text: 

Cell H13: =$H$4&$H$5&$H$6&$H$7&$H$8&$H$9&$H$10&$H$11 

The entire message is then in cell J4:   =Language!$E$156&"  "&$H$13&"." 

Formula in cell J5:   =Language!$E$157 

The message in J5 appears when there is a zero in D4. 

 

Line 7 of the “World Cup” worksheet is intended for the message. We merge cells B7:X7 into one cell and center its 

contents. 

In cell B7 we enter the following formula: 

=IF(GrpA!$K$10+GrpB!$K$10+GrpC!$K$10+GrpD!$K$10+GrpE!$K$10+GrpF!$K$10+GrpG!$K$10 

+GrpH!$K$10=0,"",IF(Distinctness!$D$4>0,Distinctness!$J$4,Distinctness!$J$5)) 

The first condition 

GrpA!$K$10+GrpB!$K$10+GrpC!$K$10+GrpD!$K$10+GrpE!$K$10+GrpF!$K$10+GrpG!$K$10+GrpH!$K$10=0 

means that no message appears if no results have yet been entered in the schedule. 

Cell B7 is given a green font. With the help of the conditional formatting, the font color changes to red if there is an 

unclear order of precedence in one of the groups. The formula for the condition is: 

=Distinctness!$D$4>0 . 

 

The current status of the schedule can be found in the file named “WCup_Tut_9.xlsx”. 

10 Optional:  Additional overview of the direct comparisons 
 (contains matrix formulas) 

This overview is intended to show the comparison table of the three teams concerned, when comparing three 

teams directly, in order to make it clear how the ranking is established in this case. The direct comparison of two 

teams is also shown in this overview. 

We create a new spreadsheet named "Direct comparisons". It should look like this: 

Columns A to V are the range visible to the user. Columns W to AG are hidden. 

 

We will develop the formulas for group A and then copy the entire range B3:AG9 seven times. In each of these co-

pies we then only need to adapt the formula for the heading (cells C11, C19 etc.) and to replace the character string 

“GrpA” with “GrpB” or “GrpC” etc. in the formulas by Find/Replace. Therefore, the setting or omission of the dollar 

signs must be carefully observed here. 

10.1 Visible area - part 1 

We start with the heading: 

Cell B1: =Language!$E$116 

The cells B1 to U1 are merged. 

For the headings in line 3 we enter the following formulas: 

Cell C3: =Language!$E$124&" A" 

Cell D3: =Language!$E$129 

Cell I3: =Language!$E$132 

Cell N3: =Language!$E$133 

For the headings in line 4 we enter the following formulas: 

Cell D4: =Language!$E$125 

Cell E4: =Language!$E$126 

Cell F4: =Language!$E$127 

Cell K4: =Language!$E$125 

Cell N4: =Language!$E$125 

Cell O4: =Language!$E$126 

Cell P4: =Language!$E$127 

Cell T4: =Language!$E$130 

Cell U4: =Language!$E$131 

We enter the current number in cells A5 to A8. We copy the team names in sorted order from the worksheet 

"GrpA": 

Cell C5: =GrpA!$O$11 

Cell C6: =GrpA!$O$12 

Cell C7: =GrpA!$O$13 

Cell C8: =GrpA!$O$14 

For example, for later copying of the complete group, it is important that there is a dollar sign in front of the num-

bers 11, 12, 13 and 14. 

In range D5:H8 you should first see a copy of the group table: 

Cell D5: =GrpA!$P$11 

Cell D6: =GrpA!$P$12 

Cell D7: =GrpA!$P$13 

Cell D8: =GrpA!$P$14 

Cell E5: =GrpA!$Q$11 

Cell E6: =GrpA!$Q$12 

Cell E7: =GrpA!$Q$13 

Cell E8: =GrpA!$Q$14 

Cell F5: =GrpA!$R$11 

Cell F6: =GrpA!$R$12 

Cell F7: =GrpA!$R$13 

Cell F8: =GrpA!$R$14 

Cell H5: =GrpA!$S$11 

Cell H6: =GrpA!$S$12 

Cell H7: =GrpA!$S$13 

Cell H8: =GrpA!$S$14 

 

In the range K5:K8 the points of a direct comparison of two teams should be displayed. If there is no direct compa-

rison between two teams, the fields remain empty. 

In order to determine whether a direct comparison with another team is taking place for the team in question, we 

first add two more formulas to the group tables. 

On the “GrpA” worksheet, we enter the formula =(P4+S4>0)  in cell X4 and copy it down to cell X7. This formula 

returns a boolean value; it enters TRUE or FALSE in the cell. TRUE means that at least one match result has been 

entered for the team in question, FALSE means that no match result has yet been entered for this team. This is ba-

sed on the following consideration: A team where both the number of points (P4) and the number of goals accep-

ted (S4) is zero cannot have played a match yet. If they have played a match but have zero points, at least one goal 

must have been accepted, otherwise they could not have lost. 

In cell J18 of the "GrpA" worksheet, we enter a formula that returns a „1“ if there is a direct two-way comparison 

for the team in question and the team has played at least one match, otherwise a „0“. The formula is: 

=IF($X$4,IF(AND($Y$4=$Y$5,$Y$4<>$Y$6,$Y$4<>$Y$7),1,0)+IF(AND($Y$4=$Y$6,$Y$4<>$Y$5,$Y$4<>$Y$7),1,0)+IF

(AND($Y$4=$Y$7,$Y$4<>$Y$5,$Y$4<>$Y$6),1,0),0) 

The structure of the formula is: 

=IF($X$4,IF(...)+IF(...)+IF(...),0) 

The outer IF function returns a „0“ if the team in question has not played a match. Otherwise, a sum of three IF 

functions is calculated. 

The first IF function IF(AND($Y$4=$Y$5,$Y$4<>$Y$6,$Y$4<>$Y$7),1,0)  checks whether the ranking of the team 

concerned ($Y$4) matches the rank in cell $Y$5, but none of the others. The other two IF functions check the other 

two possibilities for a two-way comparison. Only one of the three IF functions can ever apply. The formula therefo-

re enters a „0“ or a „1“ in cell J18. 

We copy this formula down to cell J21. 

Now we can enter the following formulas on the "Direct comparisons" worksheet in range K5:K8: 

Cell K5: 

=IF(VLOOKUP(GrpA!$O$11,GrpA!$D$18:$J$21,7,0)=0,"",VLOOKUP(GrpA!$O$11,GrpA!$D$18:$J$21,6,0)) 

Cell K6: 

=IF(VLOOKUP(GrpA!$O$12,GrpA!$D$18:$J$21,7,0)=0,"",VLOOKUP(GrpA!$O$12,GrpA!$D$18:$J$21,6,0)) 

Cell K7: 

=IF(VLOOKUP(GrpA!$O$13,GrpA!$D$18:$J$21,7,0)=0,"",VLOOKUP(GrpA!$O$13,GrpA!$D$18:$J$21,6,0)) 

Cell K8: 

=IF(VLOOKUP(GrpA!$O$14,GrpA!$D$18:$J$21,7,0)=0,"",VLOOKUP(GrpA!$O$14,GrpA!$D$18:$J$21,6,0)) 

Also important here: dollar signs before 11, 12, 13 and 14! 

The first VLOOKUP function accesses the value in column J of the “GrpA” worksheet and uses it to check whether a 

direct two-way comparison is taking place for the team in question. If so, the value of the second VLOOKUP func-

tion (the score 3, 1, or 0) is displayed; otherwise, the cell is blank. 

 

In range N5:R8, the group table should be displayed for a direct comparison of three teams. To do this, additional 

formulas must first be prepared in the hidden area (see Chapter 10.2). We will enter and discuss these formulas 

later (see Chapter 10.3). 

 

In range T5:T8 it should be shown whether a bonus for fair play/lot has already been entered: 

Cell T5: 

=SUMIFS(FairPlayPoints1,FairPlayTeams1,$C5)+SUMIFS(FairPlayPoints2,FairPlayTeams2,$C5) 

The formula is copied down to cell T8. 

 

A red dot should appear in the range U5:U8 if the positioning for the team in question is not yet clear. We will enter 

and discuss the formulas for this later (see Chapter 10.3). 

10.2 Hidden area 

In order to show the direct comparisons of three teams, we need a preparation in the hidden area 

(columns W to AG). 

Four different groups of three are possible in each group. The tables for these four groups of three can be found on 

the “GrpA” worksheet in range N16:V35. If there is no direct comparison between three teams, these four tables 

are all zeros. Only in the case of a direct comparison of three teams do numbers not equal to zero appear in one of 

the four tables. We want to copy exactly this table into the range AD5:AG7 of our worksheet "Direct comparisons". 

To do this, we need to know which are the correct line numbers. 

We enter the following formula in cell Y5: 

=IF(VLOOKUP($C5,GrpA!$O$17:$V$20,8,0)>0,MATCH($C5,GrpA!$O$17:$O$20,0),0) 

The expression VLOOKUP($C5,GrpA!$O$17:$V$20,8,0)  looks for the relevant team ($C5) on the worksheet "GrpA" 

in column V to see whether there is a zero (not a direct Comparison) or a number greater than zero (direct compa-

rison). 

The expression MATCH($C5,GrpA!$O$17:$O$20,0)  tells us at which position the respective team ($C5) is in the 

range GrpA!$O$17:$O$20 . 

The formula in Y5 therefore returns the position of the team in question within the first of the four possible groups 

of three if a direct comparison is available, otherwise a zero. 

In the same way, the formulas in cells Z5, AA5 and AB5 return the team's position in the other three possible 

groups of three. These formulas are: 

Cell Z5: =IF(VLOOKUP($C5,GrpA!$O$22:$V$25,8,0)>0,MATCH($C5,GrpA!$O$22:$O$25,0),0) 

Cell AA5: =IF(VLOOKUP($C5,GrpA!$O$27:$V$30,8,0)>0,MATCH($C5,GrpA!$O$27:$O$30,0),0) 

Cell AB5: =IF(VLOOKUP($C5,GrpA!$O$32:$V$35,8,0)>0,MATCH($C5,GrpA!$O$32:$O$35,0),0) 

We copy the formulas in cells Y5, Z5, AA5 and AB5 down to row 8. 

In line 9 we form the sum of the respective column: 

Cell Y9: =SUM(Y5:Y8) 

Cell Z9: =SUM(Z5:Z8) 

Cell AA9: =SUM(AA5:AA8) 

Cell AB9: =SUM(AB5:AB8) 

Those four numbers are then either all zero or one of them is a six. 

 

We enter the following formula in cell AC9:  {=MAX(IF($Y9:$AB9>0,SPALTE($Y9:$AB9)-SPALTE(X$1),0))} 

This is a so-called matrix formula. The curly brackets are not allowed to be entered. But you have to finish entering 

the formula with Ctrl-Shift-Enter. Then the curly brackets are set automatically. 

This matrix formula calculates the four expressions 

IF($Y9>0,COLUMN($Y9)-COLUMN(X$1),0) 

IF($Z9>0,COLUMN($Z9)-COLUMN(X$1),0) 

IF($AA9>0,COLUMN($AA9)-COLUMN(X$1),0) 

IF($AB9>0,COLUMN($AB9)-COLUMN(X$1),0) 

and returns the maximum of it. 

For example, in the following figure we see a six in cell Z9. 

The first, third and fourth expressions result in 0.  Only for the second expression, the condition  $Z9>0  is true, so 

that the expression  COLUMN($Z9)-COLUMN(W$1)  is calculated here. This expression gives 25 - 23 = 2. The maxi-

mum of the four numbers 0, 0, 2, 0 is then 2. 

The number in cell AC9 tells us that in this example, column 2 of the small table contains line numbers greater 

than 0. 

So we know that we can find the positions or the relative row numbers of the teams in column 2 of the small table 

Y5:AB8. We will use this in the formulas in the N5:R8 range. 

We also know that we have to copy the second table from the “GrpA” worksheet from the four tables in range 

P17:S35 to the “Direct comparisons” worksheet. And that's what we're going to do next. We use a matrix formula 

here again, this time not in a single cell, but in a range. 

To do this, we select the range AD5:AG7 on the "Direct comparisons" worksheet and enter the following formula in 

the editing line above: 

=IF($Y9>0,GrpA!$P$17:$S$19,IF($Z9>0,GrpA!$P$22:$S$25,IF($AA9>0,GrpA!$P$27:$S$30, 

IF($AB9>0,GrpA!$P$32:$S$35,"0")))) 

The entry must be completed with Ctrl-Shift-Enter. The formula is then automatically enclosed in curly brackets. 

The formula causes the second table in our example to be copied to the "Direct comparisons" worksheet. 

The formula consists of nested IF functions. If $Y9>0, the range  GrpA!$P$17:$S$19  should be copied. If $Z9>0, the 

range  GrpA!$P$22:$S$25  should be copied etc.  Otherwise "0" is entered everywhere. 

With the help of the matrix formula, we only need to enter a single formula for the entire range 

(the so-called matrix). 

The alternative would be to enter a formula of the type  

=IF($Y9>0,GrpA!$P$17,IF($Z9>0,GrpA!$P$22,IF($AA9>0,GrpA!$P$27,IF($AB9>0,GrpA!$P$32,"0"))))   

in everyone of the 12 cells. 

 

We want to put a red dot in cells U5 to U8 for each team with an unclear rank. In preparation for this, we first crea-

te zeros or ones in cells W5 to W8. We enter the following formulas in cells W5 to W8: 

Cell W5: 

=IF(AND(OR(TRUNC(GrpA!$T$11,4)=TRUNC(GrpA!$T$12,4),TRUNC(GrpA!$T$11,4)=TRUNC(GrpA!$T$13,4),TRUNC

(GrpA!$T$11,4)=TRUNC(GrpA!$T$14,4)),VLOOKUP(GrpA!$O$11,GrpA!$O$4:$X$7,10,0)),1,0) 

Cell W6: 

=IF(AND(OR(TRUNC(GrpA!$T$12,4)=TRUNC(GrpA!$T$11,4),TRUNC(GrpA!$T$12,4)=TRUNC(GrpA!$T$13,4),TRUNC

(GrpA!$T$12,4)=TRUNC(GrpA!$T$14,4)),VLOOKUP(GrpA!$O$12,GrpA!$O$4:$X$7,10,0)),1,0) 

Cell W7: 

=IF(AND(OR(TRUNC(GrpA!$T$13,4)=TRUNC(GrpA!$T$11,4),TRUNC(GrpA!$T$13,4)=TRUNC(GrpA!$T$12,4),TRUNC

(GrpA!$T$13,4)=TRUNC(GrpA!$T$14,4)),VLOOKUP(GrpA!$O$13,GrpA!$O$4:$X$7,10,0)),1,0) 

Cell W8: 

=IF(AND(OR(TRUNC(GrpA!$T$14,4)=TRUNC(GrpA!$T$11,4),TRUNC(GrpA!$T$14,4)=TRUNC(GrpA!$T$12,4),TRUNC

(GrpA!$T$14,4)=TRUNC(GrpA!$T$13,4)),VLOOKUP(GrpA!$O$14,GrpA!$O$4:$X$7,10,0)),1,0) 

If the ranking of the team in question, reduced to four decimal places, matches the reduced ranking of another 

team, a 1 is entered, otherwise a 0. 

The AND function AND(OR(…),VLOOKUP(GrpA!$O$14,GrpA!$O$4:$X$7,10,0)) has the effect that a „1“ is only en-

tered if at least one match has been entered for the team in question. 

The expression VLOOKUP(GrpA!$O$14,GrpA!$O$4:$X$7,10,0) ) reads a Boolean value from column X of the works-

heet "GrpA" that indicates whether the number of games entered for the relevant Crew is greater than zero. 

With these formulas, too, it is important that the numbers 11, 12, 13 and 14 are preceded by the dollar sign! 

10.3 Visible area - part 2 

In cell U5 we enter the formula  =IF($W5>0;"•";"")  and copy it down to cell U8. 

The special character can be entered in any cell (Insert  Symbol) and then inserted into the editing line with Copy 

+ Paste. 

A red dot appears for the teams with an unclear ranking. 

 

In the range N5:R8 the table for the direct comparison of three teams should be shown. 

In cell N5 we enter the following formula: 

=IF($AC9=0,"",IF(INDEX($Y5:$AB8,1,$AC9)=0,"",INDEX($AD5:$AG7,INDEX($Y5:$AB8,1,$AC9),1))) 

Explanation of the formula: 

IF($AC9=0;"" means: if there is no direct comparison of three teams, nothing is entered. 

The INDEX function in the form INDEX (range, row, column) returns the content of a cell from the specified range. 

The relative positions of the cell in this range are entered in the parameters row and column. For example, the ex-

pression INDEX(C4:H9;2;3) would return the contents of cell E5. 

If there is a direct comparison, there is a number greater than zero in cell AC9 and in the small table Y5:AB8 one of 

the four columns contains the row numbers of the three teams involved in the direct comparison. In cell AC9 is the 

number of the column in which we can find these line numbers. 

The expression IF(INDEX($Y5:$AB8,1,$AC9)=0  therefore checks whether the line number in the first line is equal to 

0. If this is the case, then it is the team that does not take part in the direct comparison, and nothing is entered. 

Otherwise, the expression INDEX($AD5:$AG7,INDEX($Y5:$AB8,1,$AC9),1) is evaluated and the result is returned. 

In this expression we have two nested INDEX functions. 

The inner INDEX function  INDEX($Y5:$AB8,1,$AC9)  gives us the relative row number of the team in question, the 

outer INDEX function gives us the score of the team in question from column 1 of table AD5:AG7. 

If we insert a “2” instead of a “1” for the column number, we get the goal difference and with a “3” or a “4” the 

goals. We use this for the formulas in cells O5, P5 and R5. They are: 

Cell O5: 

=IF($AC9=0,"",IF(INDEX($Y5:$AB8,1,$AC9)=0,"",INDEX($AD5:$AG7,INDEX($Y5:$AB8,1,$AC9),2))) 

Cell P5: 

=IF($AC9=0,"",IF(INDEX($Y5:$AB8,1,$AC9)=0,"",INDEX($AD5:$AG7,INDEX($Y5:$AB8,1,$AC9),3))) 

Cell R5: 

=IF($AC9=0,"",IF(INDEX($Y5:$AB8,1,$AC9)=0,"",INDEX($AD5:$AG7,INDEX($Y5:$AB8,1,$AC9),4))) 

The remaining formulas in table M5:Q8 all look very similar: 

Cell N6: 

=IF($AC9=0,"",IF(INDEX($Y5:$AB8,2,$AC9)=0,"",INDEX($AD5:$AG7,INDEX($Y5:$AB8,2,$AC9),1))) 

Cell O6: 

=IF($AC9=0,"",IF(INDEX($Y5:$AB8,2,$AC9)=0,"",INDEX($AD5:$AG7,INDEX($Y5:$AB8,2,$AC9),2))) 

Cell P6: 

=IF($AC9=0,"",IF(INDEX($Y5:$AB8,2,$AC9)=0,"",INDEX($AD5:$AG7,INDEX($Y5:$AB8,2,$AC9),3))) 

Cell R6: 

=IF($AC9=0,"",IF(INDEX($Y5:$AB8,2,$AC9)=0,"",INDEX($AD5:$AG7,INDEX($Y5:$AB8,2,$AC9),4))) 

Cell N7: 

=IF($AC9=0,"",IF(INDEX($Y5:$AB8,3,$AC9)=0,"",INDEX($AD5:$AG7,INDEX($Y5:$AB8,3,$AC9),1))) 

Cell O7: 

=IF($AC9=0,"",IF(INDEX($Y5:$AB8,3,$AC9)=0,"",INDEX($AD5:$AG7,INDEX($Y5:$AB8,3,$AC9),2))) 

Cell P7: 

=IF($AC9=0,"",IF(INDEX($Y5:$AB8,3,$AC9)=0,"",INDEX($AD5:$AG7,INDEX($Y5:$AB8,3,$AC9),3))) 

Cell R7: 

=IF($AC9=0,"",IF(INDEX($Y5:$AB8,3,$AC9)=0,"",INDEX($AD5:$AG7,INDEX($Y5:$AB8,3,$AC9),4))) 

Cell N8: 

=IF($AC9=0,"",IF(INDEX($Y5:$AB8,4,$AC9)=0,"",INDEX($AD5:$AG7,INDEX($Y5:$AB8,4,$AC9),1))) 

Cell O8: 

=IF($AC9=0,"",IF(INDEX($Y5:$AB8,4,$AC9)=0,"",INDEX($AD5:$AG7,INDEX($Y5:$AB8,4,$AC9),2))) 

Cell P8: 

=IF($AC9=0,"",IF(INDEX($Y5:$AB8,4,$AC9)=0,"",INDEX($AD5:$AG7,INDEX($Y5:$AB8,4,$AC9),3))) 

Cell R8: 

=IF($AC9=0,"",IF(INDEX($Y5:$AB8,4,$AC9)=0,"",INDEX($AD5:$AG7,INDEX($Y5:$AB8,4,$AC9),4))) 

 

A large blue dot should appear to the left of the group name so that the groups in which a direct comparison is car-

ried out are even more visible. In cell B3 we enter the following formula: 

=IF(OR($AC9>0,$K5<>"",$K6<>"",$K7<>""),"","") 

The special character can be entered again in any cell with Insert  Symbol and then inserted into the editing line 

with Copy + Paste. It may appear as a space in the editing line, but it still works. 

 

We now select the range B3:AG9 and copy it with Ctrl C, for example. 

Then we click on cell B11 and paste the copy, for example with Ctrl V. 

Important: The new range is still selected! 

With Find and Replace we change the text „GrpA" to „GrpB" in all formulas of the selected range. 

In cell C11 we change the formula  =Language!$E$124&" B". 

We create the ranges for the remaining groups in the same way. - All done! 

 

The final schedule can be found in the file "WCup_Tut_final.xlsx". 

11 Adjusting the schedule after the group draw 

The match schedule is designed in such a way that changes in the group allocation as well as changes in the FIFA 

schedule are easy to handle. No formulas need to be changed, just certain entries in the 'Language', 'Groups' and 

'Matches' spreadsheets. 

 

11.1 Spreadsheet 'Language'  

The list of venues will be updated here. 

There is nothing else to do on this spreadsheet. If you want, you can arrange the teams in the order of the current 

FIFA rankings. However, this is not necessary for the functionality of the schedule. Also, the numbers in column D 

may be mixed up. They just have to be unique. 

 

11.2 Allocation of the teams to the groups 

We show the worksheet 'Groups'. In column C we now enter the numbers of the teams that have been assigned to 

the group in question. For example, for team A1 we enter the number 51 because Qatar has the number 51 on the 

sheet 'Language'. For team A2 we enter the number of Ecuador, for team A3 the number of Senegal and for team 

A4 the number of the Netherlands. 

Accordingly, we enter the numbers for the remaining groups in accordance with the FIFA draw. On the spreadsheet 

‚World Cup‘ we now see the correct group participants in the correct order. 

 

11.3 Match data update 

We show the 'Matches' spreadsheet. The kick-off times must be updated in column E and the venues in column G. 

The match numbers correspond to the FIFA match numbers. If you would like to number the games differently, you 

can make the appropriate changes in column B after removing the sheet protection. 

 

11.4 Updating the match numbers on the schedule 

The match numbers are now corrected on the 'World Cup' spreadsheet. It is advisable to arrange the six group 

games within each group in chronological order. 

That's it! - All these changes are automatically applied to the rest of the spreadsheets. 

Schedule for the next World Cup tournament in 2026 

On 10 January 2017, FIFA decided in a vote that from 2026 onwards, 48 teams will take part in the final tournament 

instead of 32. 

In the group stage, instead of 8 groups of four, there should be 16 groups of three. 

After sustained criticism of the groups of three, the tournament mode was changed again on March 14, 2023. The 

48 teams now play in 12 groups of four. 

In the KO round there is also a sixteenth final with 16 KO games. The 1st and 2nd place finishers in each group and 

the 8 best third-placed teams qualify for the KO round. 

The total number of games increases from 64 to 104. 

 

So this tutorial no longer fits the new tournament schedule one-to-one, but it does fit the tournament schedule of 

the Women's World Cup. 

Source 

https://www.fifa.com/en/tournaments/mens/worldcup/qatar2022 

https://en.wikipedia.org/wiki/Time_zone 

 

https://www.fifa.com/en/tournaments/mens/worldcup/qatar2022
https://en.wikipedia.org/wiki/Time_zone

